// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.9.0) (token/ERC20/IERC20.sol)

pragma solidity ^0.8.19;

/**
 * @dev Interface of the ERC20 standard as defined in the EIP.
 */
interface IERC20 {
    /**
     * @dev Emitted when `value` tokens are moved from one account (`from`) to
     * another (`to`).
     *
     * Note that `value` may be zero.
     */
    event Transfer(address indexed from, address indexed to, uint256 value);

    /**
     * @dev Emitted when the allowance of a `spender` for an `owner` is set by
     * a call to {approve}. `value` is the new allowance.
     */
    event Approval(address indexed owner, address indexed spender, uint256 value);

    /**
     * @dev Returns the value of tokens in existence.
     */
    function totalSupply() external view returns (uint256);

    /**
     * @dev Returns the value of tokens owned by `account`.
     */
    function balanceOf(address account) external view returns (uint256);

    /**
     * @dev Moves a `value` amount of tokens from the caller's account to `to`.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transfer(address to, uint256 value) external returns (bool);

    /**
     * @dev Returns the remaining number of tokens that `spender` will be
     * allowed to spend on behalf of `owner` through {transferFrom}. This is
     * zero by default.
     *
     * This value changes when {approve} or {transferFrom} are called.
     */
    function allowance(address owner, address spender) external view returns (uint256);

    /**
     * @dev Sets a `value` amount of tokens as the allowance of `spender` over the
     * caller's tokens.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * IMPORTANT: Beware that changing an allowance with this method brings the risk
     * that someone may use both the old and the new allowance by unfortunate
     * transaction ordering. One possible solution to mitigate this race
     * condition is to first reduce the spender's allowance to 0 and set the
     * desired value afterwards:
     * https://github.com/ethereum/EIPs/issues/20#issuecomment-263524729
     *
     * Emits an {Approval} event.
     */
    function approve(address spender, uint256 value) external returns (bool);

    /**
     * @dev Moves a `value` amount of tokens from `from` to `to` using the
     * allowance mechanism. `value` is then deducted from the caller's
     * allowance.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transferFrom(address from, address to, uint256 value) external returns (bool);
}

/// @notice Simple single owner authorization mixin.
/// @author Solmate (https://github.com/transmissions11/solmate/blob/main/src/auth/Owned.sol)
abstract contract Owned {
    /*//////////////////////////////////////////////////////////////
                                 EVENTS
    //////////////////////////////////////////////////////////////*/

    event OwnershipTransferred(address indexed user, address indexed newOwner);

    /*//////////////////////////////////////////////////////////////
                            OWNERSHIP STORAGE
    //////////////////////////////////////////////////////////////*/

    address public owner;

    modifier onlyOwner() virtual {
        require(msg.sender == owner, "UNAUTHORIZED");

        _;
    }

    /*//////////////////////////////////////////////////////////////
                               CONSTRUCTOR
    //////////////////////////////////////////////////////////////*/

    constructor(address _owner) {
        owner = _owner;

        emit OwnershipTransferred(address(0), _owner);
    }

    /*//////////////////////////////////////////////////////////////
                             OWNERSHIP LOGIC
    //////////////////////////////////////////////////////////////*/

    function transferOwnership(address newOwner) public virtual onlyOwner {
        owner = newOwner;

        emit OwnershipTransferred(msg.sender, newOwner);
    }
}


contract Timelock is Owned {
    uint256 public delay;
    address public token;

    struct PendingTransaction {
        address destination;
        uint256 amount;
        uint256 dueTime;
    }

    PendingTransaction[] public transactions;

    event TransactionPrepared(uint256 indexed index, address indexed destination, uint256 amount, uint256 dueTime);
    event TransactionExecuted(address indexed destination, uint256 amount);
    event TransactionCanceled(uint256 index);

    constructor(uint256 _delay, address _token) Owned(msg.sender) {
        require(_token != address(0), "Invalid token address");
        delay = _delay;
        token = _token;
    }

    function getNextNonce() public view returns (uint256) {
        return transactions.length;
    }

    function prepareTransaction(address _destination, uint256 _amount) public onlyOwner {
        require(_destination != address(0), "Invalid destination address");

        transactions.push(PendingTransaction({
            destination: _destination,
            amount: _amount,
            dueTime: block.timestamp + delay
        }));

        emit TransactionPrepared(transactions.length - 1, _destination, _amount, block.timestamp + delay);
    }

    function cancelTransaction(uint256 index) public onlyOwner {
        require(index < transactions.length, "Invalid index");
        PendingTransaction storage txn = transactions[index];
        require(txn.amount != 0, "Transaction already canceled");

        txn.amount = 0;
        emit TransactionCanceled(index);
    }

    function executeTransaction(uint256 index) public {
        require(index < transactions.length, "Invalid index");

        PendingTransaction storage txn = transactions[index];

        require(txn.dueTime <= block.timestamp, "Transaction is not due yet");
        require(txn.amount != 0, "Transaction already executed or canceled");

        uint256 amount = txn.amount;
        address destination = txn.destination;

        delete transactions[index];

        IERC20(token).transfer(destination, amount);

        emit TransactionExecuted(destination, amount);
    }
}